


Simple GTL Example

```
proc sort data=sashelp.class out=class;
  by sex;
run;
proc template;
  define statgraph groupreg;
    begingraph;
      entrytitle 'Separate Fit by Sex';
      layout overlay;
        scatterplot y=weight x=height /
          group=sex;
        regressionplot y=weight x=height /
          group=sex degree=3 name='reg';
        discretelegend 'reg' / title='Sex';
      endlayout;
    endgraph;
  end;
run;
proc sgrender data=class template=groupreg;
run;
```



Other GTL Statements


notes *"template-description"*;
 dynamic *dynamic-variable list*;
 mvar *macro-variable list*;
 nmvar *numeric-macro-variable list*;
 if (condition) *GTL-statements*
 else *GTL-statements*
 endif;
 sidebar / <options>; *GTL-statements* endsidebar;
 cell / <options>; *GTL-statements* endcell;

Layout Lattice Example

```
proc template;
  define statgraph res;
    begingraph;
      entrytitle 'Residuals by Predictor';
      layout lattice / rows=2 columns=2;
        layout overlay;
          scatterplot y=r x=height / group=sex;
          loessplot y=r x=height;
        endlayout;
        layout overlay;
          boxplot y=r x=sex;
        endlayout;
      layout overlay;
        scatterplot y=r x=age / group=sex;
        loessplot y=r x=age;
      endlayout;
    endlayout;
  endgraph;
end;
run;
```

```
proc glm data=sashelp.class;
  class sex;
  model weight = height sex age;
  output residuals=r out=r;
quit;
```

```
proc sgrender data=r template=res;
  label r= 'Residuals';
run;
```



For more information, see the following:

Basic ODS Graphics Examples
<http://support.sas.com/documentation/prod-p/grstat/9.4/en/PDF/odsbasicg.pdf>

Advanced ODS Graphics Examples
<http://support.sas.com/documentation/prod-p/grstat/9.4/en/PDF/odsadvvg.pdf>

For complete information, see the SAS 9.4 documentation at
<http://support.sas.com/documentation/>.



Graph Template Language Tip Sheet

This tip sheet collects frequently used information in one place so you don't have to search through the online documentation. It also gives you examples to take home and try.

ODS Graphics is an extension of ODS (the Output Delivery System). The Graph Template Language (GTL) specifies the layout and details of each graph produced by ODS. This powerful language includes statements for specifying plot layouts (such as lattices and overlays), plot types (such as scatter plots and histograms), and text elements (such as titles, footnotes, and insets). It also provides support for built-in computations (such as histogram binning) and the evaluation of expressions. Visual attributes of graphs are determined by the active ODS style. However, options are available for specifying colors, markers, and other plot features.

This tip sheet presents the statements and options that are most commonly used with the Graph Template Language in SAS[®] 9.4.

SAS⁹ Graph Template Language Tip Sheet

GTL Overview

For every graph and style, SAS provides an ODS template, which is a SAS program that gives instructions for creating the graph or style. You do not need to know *anything* about templates to create statistical graphics. With just a little knowledge of the template languages, you can modify graph and style templates and make permanent changes that apply every time you run a procedure. Some GTL statements and options are listed on this page.

Graph Template Structure

```
proc template;
  define statgraph template-name;
    begingraph / <designheight=h>
      designwidth=w> <options>;
    GTL-global-statements
    GTL-layout-block
  endgraph;
end;
run;
```

Legend Statements

```
continuouslegend names / <options>;
discretelegend names / <options>;
legenditem type=type name="name" / <options>;
mergedlegend names / <options>;
```

Text Statements

```
entry text-items / <options>;
entryfootnote text-items / <options>;
entrytitle text-items / <options>;
```

Linear Axis Options

```
integer=boolean
minorticks=boolean
origin=number
thresholdmax=number
thresholdmin=number
tickvaluefitpolicy=thin | rotate | stagger / other
tickvalueformat=(format-options) | data | format
tickvaluelist=(numeric-list)
tickvaluepriority=boolean
tickvaluesequence=(start=s end=e increment=i)
viewmax=number
viewmin=number
```

Layout Statements

```
layout datalattice rowvar=r-var columnvar=c-var
  / <options>;
layout datapanel classvars = (c-vars) / <options>;
layout gridded / <options>;
layout lattice / <columns=nc> <rows=nr>
  <rowdatarange=data | union | unionall>
  <columdatarange=data | union | unionall>
  <rowweights=(list)> <columnweights=(list)>
  <rowgutter=rg> <columngutter=cg> <options>;
layout overlay / <xaxisopts=(axis-options)>
  <yaxisopts=(axis-options)>
  <x2axisopts=(axis-options)>
  <y2axisopts=(axis-options)> <options>;
layout overlayequated /
  <equatetype=square | fit | equate | squaredata>
  <commonaxisopts=(axis-options)>
  <xaxisopts=(axis-options)>
  <yaxisopts=(axis-options)> <options>;
layout overlay3d / <options>;
layout prototype / <options>;
layout region / <options>;
```

Axis Options

```
discreteopts=(discrete-axis-options)
display=standard | all | none | (display-options)
displaysecondary=none | all | standard |
  (display-options)
gridattrs=<style-element> <(line-options)>
griddisplay=auto_off | auto_on | on | off
label="string"
labelattrs=<style-element> <(text-options)>
labelsplitchar="character-list"
labelsplitjustify=justification
linearopts=(linear-axis-options)
lineextent=full | data | number
logopts=(log-axis-options)
offsetmax=auto | autocompress | number
offsetmin=auto | autocompress | number
reverse=boolean
shortlabel="string"
tickstyle=outside | inside | across
tickvalueattrs=<style-element> <(options)>
tickvaluealign=left | center | right
tickvaluealignh=top | center | bottom
timeopts=(time-axis-options)
type=auto | discrete | linear | time | log
```

Plot Statements

```
axistable x=x-var value=v-var / <options>;
bandplot y|x=var limitupper=u-var
  limitlower=l-var /
  <outlineattrs=<style-element>(line-options)>
  <options>;
barchart category=x-var <response=r-var> /
  <stat=freq | pct | sum | mean | proportion>
  <group=g-var> <orient=vertical | horizontal>
  <options>;
barchartparm category=x-var response=r-var /
  <group=g-var> <orient=vertical | horizontal>
  <options>;
bihistogram3dparm y=y-var x=x-var z=z-vars /
  <options>;
blockplot x=x-var block=b-var / <options>;
boxplot y=y-var <x=x-var> /
  <orient=vertical | horizontal> <options>;
boxplotparm y=y-var <x=x-var> stat=s-var /
  <orient=vertical | horizontal> <options>;
bubbleplot y=y-var x=x-var size=size-var /
  <bubblerradiusmax=dimension>
  <bubblerradiusmin=dimension> <options>;
contourplotparm y=y-var x=x-var z=z-vars /
  <options>;
dendrogram nodeid=n-var parentid=p-var
  clusterheight=c-var / <options>;
densityplot var / <kernel(options)>
  <normal(options)> <options>;
dropline y=y-var x=x-var / dropto=x | y | both
  <label=l-var> <options>;
ellipse y=y-var x=x-var / <options>;
ellipseparm xorigin=x yorigin=y
  semimajor=s1-var semiminor=s2-var
  slope=slope / <options>;
fringeplot var / <options>;
heatmap<parm> y=y-var x=x-var /
  <colormodel=(color-list)> <options>;
highlowplot y | y=var low=l-var high=h-var /
  <options>;
histogram var / <orient=vertical | horizontal>
  <options>;
histogramparm y=y-var x=x-var /
  <orient=vertical | horizontal> <options>;
linechart category=x-var <response=r-var> /
  <options>;
lineparm y=y-var x=x-var slope=s /
  <group=g-var> <options>;
```

Plot Statements

```
loessplot y=y-var x=x-var / <alpha=a>
  <clm="name"> <degree=1|2> <smooth=auto|s>
  <interpolation=linear | cubic> <maxpoints=m>
  <lineattrs=<style-element>(line-options)>
  <group=g-var> <options>;
modelband name / <options>;
mosaicplotparm category=vars count=c-var /
  <options>;
needleplot y=y-var x=x-var / <group=g-var>
  <options>;
pbsplineplot y=y-var x=x-var / <alpha=a>
  <cli="name"> <clm="name"> <degree=d>
  <freq=f-var> <maxpoints=m> <nknots=n>
  <smooth=auto | s> <weight=w-var> <options>;
polygonplot y=y-var x=x-var id=i-var / <options>;
referenceline x | y=var / <options>
  <lineattrs=<style-element>(line-options)>;
regressionplot y=y-var x=x-var / <alpha=a>
  <cli="name"> <clm="name"> <degree=d>
  <freq=f-var> <maxpoints=m> <weight=w-var>
  <lineattrs=<style-element>(line-options)>
  <group=g-var> <options>;
scatterplot y=y-var x=x-var / <datalabel=l-var>
  <group=g-var> <markercharacter=m-var>
  <markerattrs=<style-element>(line-options)>
  <options>;
scatterplotmatrix vars / <datalabel=l-var>
  <group=g-var> <markercharacter=m-var>
  <diagonal=(<histogram> <normal> <kernel>)>
  <markerattrs=<style-element>(line-options)>
  <options>;
seriesplot y=y-var x=x-var / <group=g-var>
  <display=standard | all>
  <lineattrs=<style-element>(line-options)>
  <options>;
stepplot y=y-var x=x-var / <datalabel=l-var>
  <lineattrs=<style-element>(line-options)>
  <markerattrs=<style-element>(line-options)>
  <group=g-var> <options>;
surfaceplotparm y=y-var x=x-var z=z-vars /
  <options>;
textplot y=y-var x=x-var text=t-var / <options>;
vectorplot y=y-var x=x-var xorigin=0 yorigin=0 /
  <datalabel=l-var> <group=g-var>
  <lineattrs=<style-element>(line-options)>
  <options>;
waterfallchart category=x-var <response=r-var> /
  <options>;
```